The HOPF MAP

Maps S3 to S2

R. M. Kiehn

June 15,1998

```
> restart: with(linalg):with(difforms):with(liesymm):with(plots):
```

Warning, new definition for norm

```
Warning, new definition for norm
Warning, new definition for trace
Warning, new definition for `&^`
Warning, new definition for close
Warning, new definition for d
Warning, new definition for mixpar
Warning, new definition for wdegree
```

Cartan's Repere Mobile will be used for a representation of the Hopf Map from S3 to S2 in 4 space.

The Position Vector R4 in 4 space will be presumed to have components {X,Y,Z,S}.

The position vector r3 in 3 space will be presumed to have components {x,y,z}.

The origins are presumed to be equal, in the sense that the four sphere goes to zero when the 3 sphere goes to zero.

```
> setup(X,Y,Z,S):
```

```
> R4:=((S)^2+X^2+Y^2+Z^2)^(1/2);r3:=(x^2+y^2+z^2)^(1/2);
```

$$R4 := \sqrt{S^2 + X^2 + Y^2 + Z^2}$$

$$r3 := \sqrt{x^2 + y^2 + z^2}$$

beta is an arbitrary function on the Holder type with constant coefficients a,b,c,e.

```
> beta:=zeta*((e*S^p+a*X^p+b*Z^p+c*Y^p)^(n/p));
```

$$\beta := \zeta\,(e\,S^p + a\,X^p + b\,Z^p + c\,Y^p)^{\left(\frac{n}{p}\right)}$$

The 4 dimensional system is presumed to be represented by a basis of 4 independent 1-forms, {e1,e2,e3,e4}. The components of each 1-form can be used to form a basis frame for vectors at any point in R4, {E1,E2,E3,E4}. Each vector in the basis can be multiplied by an arbitrary factor.

The Hopf map is defined as as a projection from R4 to R3, by means of the functions:

There are two case (left and right handed) for which the Chiral factor is plus or minus 1. The choice can be changed, but is fixed on the next line.

```
> ch:=-1;
```

$$ch := -1$$

```
> x:=2*(S*Y+ch*X*Z);e1:=d(x);E1:=[ch*Z,S,ch*X,Y];
```

$$x := 2\,S\,Y - 2\,X\,Z$$

$$e1 := 2\,\mathrm{d}(S)\,Y + 2\,S\,\mathrm{d}(Y) - 2\,\mathrm{d}(X)\,Z - 2\,X\,\mathrm{d}(Z)$$

$$E1 := [-Z, S, -X, Y]$$

```
> y:=2*(Y*Z-ch*S*X);e2:=d(y);E2:=[-ch*S,Z,Y,-ch*X];
```

$$y := 2\,Y\,Z + 2\,S\,X$$

$$e2 := 2\,\mathrm{d}(Y)\,Z + 2\,Y\,\mathrm{d}(Z) + 2\,\mathrm{d}(S)\,X + 2\,S\,\mathrm{d}(X)$$

$$E2 := [S, Z, Y, X]$$

```
> z:=(X^2+Y^2)-(((S)^2+Z^2));e3:=d(z);E3:=[X,Y,-Z,-S];factor((r3^2))
  ;
```

$$z := X^2 + Y^2 - S^2 - Z^2$$

$$e3 := 2\,X\,\mathrm{d}(X) + 2\,Y\,\mathrm{d}(Y) - 2\,S\,\mathrm{d}(S) - 2\,Z\,\mathrm{d}(Z)$$

$$E3 := [X, Y, -Z, -S]$$

$$(S^2 + X^2 + Y^2 + Z^2)^2$$

The fourth 1-form will not be described by the map, but may be computed from the Adjoint to the three 1-forms, defined as e1^e2^e3.

```
> Adjoint:=subs(wcollect(factor(wcollect(e1&^e2&^e3)))):A1:=factor(g
  etcoeff(Adjoint&^d(X)));A2:=factor(getcoeff(Adjoint&^d(Y)));A3:=fa
  ctor(getcoeff(Adjoint&^d(Z)));A4:=factor(getcoeff(Adjoint&^d(S)));
>
```

$$A1 := 8\,(S^2 + X^2 + Y^2 + Z^2)\,Y$$

$$A2 := -8\,(S^2 + X^2 + Y^2 + Z^2)\,X$$

$$A3 := -8\,(S^2 + X^2 + Y^2 + Z^2)\,S$$

$$A4 := 8\,(S^2 + X^2 + Y^2 + Z^2)\,Z$$

There is also a lefthanded mode

The only difference is to change the sign of the A3 and A4 terms.

To within an arbitrary factor, the components of the adjoint vector are:

```
> Nright:=[A1,A2,A3,A4]/8;
```

$$Nright := [(S^2 + X^2 + Y^2 + Z^2)\,Y, -(S^2 + X^2 + Y^2 + Z^2)\,X, -(S^2 + X^2 + Y^2 + Z^2)\,S, (S^2 + X^2 + Y^2 + Z^2)\,Z]$$

The 1-form with components constructed from the adjoint is not integrable.

```
> Adjoint_action:=factor(innerprod(Nright,[d(X),d(Y),d(Z),d(S)]));
```

$$Adjoint\_action := (S^2 + X^2 + Y^2 + Z^2)\,(-X\,\mathrm{d}(Y) - S\,\mathrm{d}(Z) + Z\,\mathrm{d}(S) + Y\,\mathrm{d}(X))$$

For purposed herein this 1-form will be scaled by the function beta, and will be used to define the righthanded Action 1-form. The lefthanded Action 1-form changes the sign of the terms dZ and dS, through the chirality coefficient, ch = plus or minus 1.

```
> Action:=wcollect(factor((Adjoint_action/(beta))));
```

$$Action := \frac{(S^2 + X^2 + Y^2 + Z^2)\,Y\,\mathrm{d}(X)}{\zeta\,\%1} - \frac{(S^2 + X^2 + Y^2 + Z^2)\,X\,\mathrm{d}(Y)}{\zeta\,\%1} - \frac{(S^2 + X^2 + Y^2 + Z^2)\,S\,\mathrm{d}(Z)}{\zeta\,\%1} + \frac{(S^2 + X^2 + Y^2 + Z^2)\,Z\,\mathrm{d}(S)}{\zeta\,\%1}$$

$$\%1 := (e\,S^p + a\,X^p + b\,Z^p + c\,Y^p)^{\left(\frac{n}{p}\right)}$$

It is apparent that the associated vectors are orthogonal to the adjoint field.

```
> factor(innerprod(E1,Nright));factor(innerprod(E2,Nright));factor(innerprod(E3,Nright));
>
```

$$0$$
$$0$$
$$0$$

Now use the (non-integrable) 1-form Action, and compute its Pfaff sequence, A, F=dA,A^F,F^F. Note that the arbitrary factor 1/beta can be adjusted such that the Pfaff dimension of the 1-form is 3, for any anisotropy coefficients and any exponent p, as long as n = 4. This means that there exists a hierarchy of conserved Torsion (topological) invariants, depending on p and the anisotropy coefficients.. Note that the Pfaff dimension of the 1-form Action is 2, if n is not equal to 2.
Hence the 1-form usually defines a symplectic manifold, but when n = 2 , the 1-form defines a contact manifold.

```
> Action;
```

$$\frac{(S^2+X^2+Y^2+Z^2)\,Y\,\mathrm{d}(X)}{\zeta\ \%1}-\frac{(S^2+X^2+Y^2+Z^2)\,X\,\mathrm{d}(Y)}{\zeta\ \%1}-\frac{(S^2+X^2+Y^2+Z^2)\,S\,\mathrm{d}(Z)}{\zeta\ \%1}+\frac{(S^2+X^2+Y^2+Z^2)\,Z\,\mathrm{d}(S)}{\zeta\ \%1}$$

$$\%1 := (e\,S^p+a\,X^p+b\,Z^p+c\,Y^p)^{\left(\frac{n}{p}\right)}$$

Next compute the 2-form for the field intensities (which satisfy the Maxwell-Faraday equations)

```
>
> Fields:=wcollect(d(Action)):BX:=factor(getcoeff(Fields&^d(X)&^d(S)));BY:=factor(getcoeff(Fields&^d(Y)&^d(S)));BZ:=factor(getcoeff(Fields&^d(Z)&^d(S)));EX:=factor(getcoeff(Fields&^d(Y)&^d(Z)));EY:=factor(getcoeff(Fields&^d(Z)&^d(X)));EZ:=factor(getcoeff(Fields&^d(X)&^d(Y)));
```

$$BX := (-2\,S\,Z\,Y^2\,e\,S^p-2\,S\,Z\,Y^2\,a\,X^p-2\,S\,Z\,Y^2\,b\,Z^p-2\,S\,Z\,Y^2\,c\,Y^p+S^3\,Z\,n\,c\,Y^p+S\,Z\,n\,c\,Y^p\,X^2+S\,Z\,n\,c\,Y^p\,Y^2+S\,Z^3\,n\,c\,Y^p+2\,X\,Y\,Z^2\,e\,S^p+2\,X\,Y\,Z^2\,a\,X^p+2\,X\,Y\,Z^2\,b\,Z^p+2\,X\,Y\,Z^2\,c\,Y^p-X\,Y\,n\,S^2\,b\,Z^p-X^3\,Y\,n\,b\,Z^p-X\,Y^3\,n\,b\,Z^p-X\,Y\,n\,Z^2\,b\,Z^p)\Big/\left(\zeta\,(e\,S^p+a\,X^p+b\,Z^p+c\,Y^p)^{\left(\frac{n}{p}\right)}\,Y\,(e\,S^p+a\,X^p+b\,Z^p+c\,Y^p)\,Z\right)$$

$$BY := (2\,X\,Y\,Z^2\,e\,S^p+2\,X\,Y\,Z^2\,a\,X^p+2\,X\,Y\,Z^2\,b\,Z^p+2\,X\,Y\,Z^2\,c\,Y^p-X\,Y\,n\,S^2\,b\,Z^p-X^3\,Y\,n\,b\,Z^p-X\,Y^3\,n\,b\,Z^p-X\,Y\,n\,Z^2\,b\,Z^p+2\,S\,Z\,X^2\,e\,S^p+2\,S\,Z\,X^2\,a\,X^p+2\,S\,Z\,X^2\,b\,Z^p+2\,S\,Z\,c\,Y^p\,X^2$$

$$-S^3 Z n a X^p - S Z n X^2 a X^p - S Z n Y^2 a X^p - S Z^3 n a X^p) \Big/ \left(\zeta (e S^p + a X^p + b Z^p + c Y^p)^{\left(\frac{n}{p}\right)} Z\right.$$
$$\left.(e S^p + a X^p + b Z^p + c Y^p) X\right)$$

$$BZ := (-2 Z^2 c Y^p - 2 Z^2 b Z^p - 2 Z^2 a X^p - 2 Z^2 e S^p - 4 X^2 c Y^p - 4 X^2 b Z^p - 4 X^2 a X^p - 4 X^2 e S^p$$
$$- 4 Y^2 c Y^p - 4 Y^2 b Z^p - 4 Y^2 a X^p - 4 Y^2 e S^p - 2 S^2 e S^p - 2 S^2 a X^p - 2 S^2 b Z^p - 2 S^2 c Y^p$$
$$+ n c Y^p Z^2 + n c Y^p X^2 + n c Y^p Y^2 + n c Y^p S^2 + n a X^p Z^2 + n a X^p Y^2 + n a X^p X^2 + n a X^p S^2) \Big/ \left(\right.$$
$$\left.\zeta (e S^p + a X^p + b Z^p + c Y^p)^{\left(\frac{n}{p}\right)} (e S^p + a X^p + b Z^p + c Y^p)\right)$$

$$EX := (2 S Z X^2 e S^p + 2 S Z X^2 a X^p + 2 S Z X^2 b Z^p + 2 S Z c Y^p X^2 - S^3 Z n a X^p - S Z n X^2 a X^p$$
$$- S Z n Y^2 a X^p - S Z^3 n a X^p - 2 X Y S^2 e S^p - 2 X Y S^2 a X^p - 2 X Y S^2 b Z^p - 2 X Y S^2 c Y^p$$
$$+ X Y n S^2 e S^p + X^3 Y n e S^p + X Y^3 n e S^p + X Y n Z^2 e S^p) \Big/ \left(\zeta (e S^p + a X^p + b Z^p + c Y^p)^{\left(\frac{n}{p}\right)} X\right.$$
$$\left.(e S^p + a X^p + b Z^p + c Y^p) S\right)$$

$$EY := -(-2 X Y S^2 e S^p - 2 X Y S^2 a X^p - 2 X Y S^2 b Z^p - 2 X Y S^2 c Y^p + X Y n S^2 e S^p$$
$$+ X^3 Y n e S^p + X Y^3 n e S^p + X Y n Z^2 e S^p - 2 S Z Y^2 e S^p - 2 S Z Y^2 a X^p - 2 S Z Y^2 b Z^p$$
$$- 2 S Z Y^2 c Y^p + S^3 Z n c Y^p + S Z n c Y^p X^2 + S Z n c Y^p Y^2 + S Z^3 n c Y^p) \Big/ \left(\zeta\right.$$
$$\left.(e S^p + a X^p + b Z^p + c Y^p)^{\left(\frac{n}{p}\right)} S (e S^p + a X^p + b Z^p + c Y^p) Y\right)$$

$$EZ := (2 Y^2 c Y^p + 2 Y^2 b Z^p + 2 Y^2 a X^p + 2 Y^2 e S^p + 2 X^2 c Y^p + 2 X^2 b Z^p + 2 X^2 a X^p + 2 X^2 e S^p$$
$$+ 4 S^2 e S^p + 4 S^2 a X^p + 4 S^2 b Z^p + 4 S^2 c Y^p + 4 Z^2 e S^p + 4 Z^2 a X^p + 4 Z^2 b Z^p + 4 Z^2 c Y^p$$
$$- n S^2 b Z^p - n X^2 e S^p - n X^2 b Z^p - n Y^2 e S^p - n Y^2 b Z^p - n Z^2 e S^p - n Z^2 b Z^p - n S^2 e S^p) \Big/ \left(\zeta\right.$$
$$\left.(e S^p + a X^p + b Z^p + c Y^p)^{\left(\frac{n}{p}\right)} (e S^p + a X^p + b Z^p + c Y^p)\right)$$

```
> BFIELD:=[BX,BY,BZ]:EFIELD:=[EX,EY,EZ]:Poincare1:=factor(subs(ch=-1,a=1,b=1,c=1,e=1,p=2,(innerprod(BFIELD,BFIELD)-innerprod(EFIELD,EFIELD))));Poincare2:=factor(innerprod(EFIELD,BFIELD));
```

$$Poincare1 := \frac{(n-2)(n-6)(X^2 + Y^2 - S^2 - Z^2)(S^2 + X^2 + Y^2 + Z^2)}{\left((S^2 + X^2 + Y^2 + Z^2)^{(1/2\, n)}\right)^2 \zeta^2}$$

$$Poincare2 := 2\,\frac{(S^2+X^2+Y^2+Z^2)^2\,(-4+n)}{\left((e\,S^p+a\,X^p+b\,Z^p+c\,Y^p)^{\left(\frac{n}{p}\right)}\right)^2\,\zeta^2}$$

In the above equations, special choice has been made for the exponent p . Moreover, the assumption has been made that the system in R4 is isotropic. The vanishing of the second Poincare invariant implies the closed integrals of the 3-form of Spin are deformation invariants.

Next compute the three form of Topological Torsion, any signature, any exponents. Note that the components of this 3-form are proportional to the position vector R4 contracted with the 4D volume element, dX^dY^dZ^dS. The 3-form of Torsion current is in the direction of the Position vector in R4. The result is independent from the exponents and the isotropy signatures!!! The sign depends upon the chirality factor.

```
> H:=wcollect(factor(factor(Action&^d(Action))));
```

$$H := 2\,\frac{Y\,(S^2+X^2+Y^2+Z^2)^2\ \&\hat{}(\mathrm{d}(X),\mathrm{d}(Z),\mathrm{d}(S))}{\%1^2\,\zeta^2} - 2\,\frac{Z\,(S^2+X^2+Y^2+Z^2)^2\ \&\hat{}(\mathrm{d}(X),\mathrm{d}(Y),\mathrm{d}(S))}{\%1^2\,\zeta^2} + 2\,\frac{S\,(S^2+X^2+Y^2+Z^2)^2\ \&\hat{}(\mathrm{d}(X),\mathrm{d}(Y),\mathrm{d}(Z))}{\%1^2\,\zeta^2} - 2\,\frac{X\ \&\hat{}(\mathrm{d}(Y),\mathrm{d}(Z),\mathrm{d}(S))\,(S^2+X^2+Y^2+Z^2)^2}{\%1^2\,\zeta^2}$$

$$\%1 := (e\,S^p+a\,X^p+b\,Z^p+c\,Y^p)^{\left(\frac{n}{p}\right)}$$

The components of the 3-form H are orthogonal to the components of the 1-form of Action (transversal). The components of the three form H are a current in the direction of the "radial expansion" position vector in R4.

Finally compute the Topological Parity 4-form

```
> K:=factor(d(Action)&^d(Action));
```

$$K := 4\,\frac{\&\hat{}(\mathrm{d}(X),\mathrm{d}(Y),\mathrm{d}(Z),\mathrm{d}(S))\,(S^2+X^2+Y^2+Z^2)^2\,(-4+n)}{\left((e\,S^p+a\,X^p+b\,Z^p+c\,Y^p)^{\left(\frac{n}{p}\right)}\right)^2\,\zeta^2}$$

Note that a specific choice of n will cause the 4-form to vanish, such that the Pffaf dimension of the Action is 3, not 4.
The 1-form then defines a contact manifold, for which there exists a unique extremal vector, such that

both the closed integrals of the Action 1-form and the Torsion 3 -form are deformation invariants. It is remarkable that this result is valid for any istropy denominator and any value of p.

In the example above, the effect of 4D radial expansions is null for n=4. The Lie derivative of the forms of the Pfaff sequence generated by A,dA,A^dA,F^F are zero in the direction of H (the torsion vector is the radial expansion vector in 4D when n=4 any p ,etc.)

Form at the point R4 the matrix array of E1, E2, E3, Action, to create Cartan's Repere Mobile.

```
> NADJ:=[Nright[1]/R4^2,Nright[2]/R4^2,Nright[3]/R4^2,Nright[4]/R4^2
  ];
```

$$NADJ := [Y, -X, -S, Z]$$

Having established linear independence of the basis frame, it is possible to introduce a scale factor that will simplify the algebra. Indeed, divide each basis vector by R4^2, the length of the 4R position vector:

```
> FT:=evalm((array([E1,E2,E3,NADJ]))/R4^2);DET:=factor(det(FT));
```

$$FT := \begin{bmatrix} -\%1 & \%2 & -\%3 & \%4 \\ \%2 & \%1 & \%4 & \%3 \\ \%3 & \%4 & -\%1 & -\%2 \\ \%4 & -\%3 & -\%2 & \%1 \end{bmatrix}$$

$$\%1 := \frac{Z}{S^2 + X^2 + Y^2 + Z^2}$$

$$\%2 := \frac{S}{S^2 + X^2 + Y^2 + Z^2}$$

$$\%3 := \frac{X}{S^2 + X^2 + Y^2 + Z^2}$$

$$\%4 := \frac{Y}{S^2 + X^2 + Y^2 + Z^2}$$

$$DET := \frac{1}{(S^2 + X^2 + Y^2 + Z^2)^2}$$

The determinant never vanishes except at the origin of R4.
Next create the induced metric:

```
> GUN:=subs(innerprod(FT,transpose(FT)));
```

$$GUN := \begin{bmatrix} \%1 & 0 & 0 & 0 \\ 0 & \%1 & 0 & 0 \\ 0 & 0 & \%1 & 0 \\ 0 & 0 & 0 & \%1 \end{bmatrix}$$

$$\%1 := \frac{1}{S^2 + X^2 + Y^2 + Z^2}$$

Note that the induce metric is conformal with a factor = to 1/R4^4

>

```
> FF:=transpose(FT);
```

$$FF := \begin{bmatrix} -\%1 & \%2 & \%3 & \%4 \\ \%2 & \%1 & \%4 & -\%3 \\ -\%3 & \%4 & -\%1 & -\%2 \\ \%4 & \%3 & -\%2 & \%1 \end{bmatrix}$$

$$\%1 := \frac{Z}{S^2 + X^2 + Y^2 + Z^2}$$

$$\%2 := \frac{S}{S^2 + X^2 + Y^2 + Z^2}$$

$$\%3 := \frac{X}{S^2 + X^2 + Y^2 + Z^2}$$

$$\%4 := \frac{Y}{S^2 + X^2 + Y^2 + Z^2}$$

```
> FFINV:=inverse(FF);
```

$$FFINV := \begin{bmatrix} -Z & S & -X & Y \\ S & Z & Y & X \\ X & Y & -Z & -S \\ Y & -X & -S & Z \end{bmatrix}$$

```
> innerprod(FF,FFINV);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> DR:=innerprod(FFINV,[d(X),d(Y),d(Z),d(S)]):sigma1:=wcollect(factor
(wcollect(DR[1])));sigma2:=wcollect(factor(wcollect(DR[2])));sigma
3:=factor(wcollect(DR[3]));omega:=factor(wcollect(DR[4]));dsigma1:
=d(sigma1);dsigma2:=d(sigma2);dsigma3:=d(sigma3);domega:=d(omega);
```

The rescaling of all the basis elements by the non-negative function yields a simple algebra:

$$\sigma 1 := -\mathrm{d}(X)\, Z + S\, \mathrm{d}(Y) - X\, \mathrm{d}(Z) + \mathrm{d}(S)\, Y$$

$$\sigma 2 := S\, \mathrm{d}(X) + \mathrm{d}(Y)\, Z + Y\, \mathrm{d}(Z) + \mathrm{d}(S)\, X$$

$$\sigma 3 := X\, \mathrm{d}(X) + Y\, \mathrm{d}(Y) - Z\, \mathrm{d}(Z) - S\, \mathrm{d}(S)$$

$$\omega := -X\, \mathrm{d}(Y) - S\, \mathrm{d}(Z) + Z\, \mathrm{d}(S) + Y\, \mathrm{d}(X)$$

$$dsigma1 := 0$$

$$dsigma2 := 0$$

$$dsigma3 := 0$$

$$domega := -2\,((\mathrm{d}(X))\,\&\hat{}\,(\mathrm{d}(Y))) + 2\,((\mathrm{d}(Z))\,\&\hat{}\,(\mathrm{d}(S)))$$

The forms sigma (1,2,3) and omega are the 1-forms relative to the basis frame FF
In the example, the 1-forms sigma1 sigma2 and sigma3 are exact differentials equal to the differentials of the mapping functions defined by the Hopf map giving x,y,z in terms of X,Y,Z,T
The non-exact and non-integrable 1-form of Action is proportional to the 1-form small omega.
For a parametrized surface, small omega vanishes, and such subspaces have Zero Torsion of the Affine type.
From the Frame use the standard methods to compute the Cartan Matrix of connection 1-forms.

See http://www.uh.edu/~rkiehn/pdf/defects2.pdf
for details of the Cartan method for an arbitrary Repere Mobile.

```
> dFF:=array([[d(FF[1,1]),d(FF[1,2]),d(FF[1,3]),d(FF[1,4])],[d(FF[2,1]),d(FF[2,2]),d(FF[2,3]),d(FF[2,4])],[d(FF[3,1]),d(FF[3,2]),d(FF[3,3]),d(FF[3,4])],[d(FF[4,1]),d(FF[4,2]),d(FF[4,3]),d(FF[4,4])]]):
> cartan:=(evalm(FFINV&*dFF)):
```

Evaluate each component of the connection coefficients on transverse space of E1,E2,E3.

```
> Gamma11:=factor(wcollect(cartan[1,1]));
```

$$\Gamma 11 := -\frac{X\,\mathrm{d}(X) + Y\,\mathrm{d}(Y) + Z\,\mathrm{d}(Z) + S\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma12:=factor(wcollect(cartan[1,2]));
```

$$\Gamma 12 := -\frac{-Y\,\mathrm{d}(X) + X\,\mathrm{d}(Y) - S\,\mathrm{d}(Z) + Z\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma13:=factor(wcollect(cartan[1,3]));
```

$$\Gamma 13 := \frac{-\mathrm{d}(X)\,Z + S\,\mathrm{d}(Y) + X\,\mathrm{d}(Z) - \mathrm{d}(S)\,Y}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma21:=factor(wcollect(cartan[2,1]));
```

$$\Gamma 21 := \frac{-Y\,\mathrm{d}(X) + X\,\mathrm{d}(Y) - S\,\mathrm{d}(Z) + Z\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma22:=factor(wcollect(cartan[2,2]));
```

$$\Gamma 22 := -\frac{X\,\mathrm{d}(X) + Y\,\mathrm{d}(Y) + Z\,\mathrm{d}(Z) + S\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma23:=factor(wcollect(cartan[2,3]));
```

$$\Gamma 23 := -\frac{-S\,\mathrm{d}(X) - \mathrm{d}(Y)\,Z + Y\,\mathrm{d}(Z) + \mathrm{d}(S)\,X}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma31:=factor(wcollect(cartan[3,1]));
```

$$\Gamma 31 := -\frac{-\mathrm{d}(X)\,Z + S\,\mathrm{d}(Y) + X\,\mathrm{d}(Z) - \mathrm{d}(S)\,Y}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma32:=factor(wcollect(cartan[3,2]));
```

$$\Gamma 32 := \frac{-S\,\mathrm{d}(X) - \mathrm{d}(Y)\,Z + Y\,\mathrm{d}(Z) + \mathrm{d}(S)\,X}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Gamma33:=factor(wcollect(cartan[3,3]));
```

$$\Gamma 33 := -\frac{X\,\mathrm{d}(X) + Y\,\mathrm{d}(Y) + Z\,\mathrm{d}(Z) + S\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

The "Space-S" components are:

```
> hh1:=factor(wcollect(cartan[4,1]));
>
```

$$hh1 := -\frac{-S\,\mathrm{d}(X) - \mathrm{d}(Y)\,Z + Y\,\mathrm{d}(Z) + \mathrm{d}(S)\,X}{S^2 + X^2 + Y^2 + Z^2}$$

```
> gg1:=factor(wcollect(factor(wcollect(cartan[1,4]))));
```

$$gg1 := \frac{-S\,\mathrm{d}(X) - \mathrm{d}(Y)\,Z + Y\,\mathrm{d}(Z) + \mathrm{d}(S)\,X}{S^2 + X^2 + Y^2 + Z^2}$$

```
> hh2:=factor(wcollect(cartan[4,2]));
```

$$hh2 := -\frac{-\mathrm{d}(X)\,Z + S\,\mathrm{d}(Y) + X\,\mathrm{d}(Z) - \mathrm{d}(S)\,Y}{S^2 + X^2 + Y^2 + Z^2}$$

```
> gg2:=factor(wcollect(cartan[2,4]));
```

$$gg2 := \frac{-\mathrm{d}(X)\,Z + S\,\mathrm{d}(Y) + X\,\mathrm{d}(Z) - \mathrm{d}(S)\,Y}{S^2 + X^2 + Y^2 + Z^2}$$

```
> hh3:=factor(wcollect(cartan[4,3]));
```

$$hh3 := -\frac{-Y\,\mathrm{d}(X) + X\,\mathrm{d}(Y) - S\,\mathrm{d}(Z) + Z\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

```
> gg3:=factor(wcollect(cartan[3,4]));
```

$$gg3 := \frac{-Y\,\mathrm{d}(X) + X\,\mathrm{d}(Y) - S\,\mathrm{d}(Z) + Z\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

```
> Omega:=factor(simplify(wcollect(cartan[4,4])));
```

$$\Omega := -\frac{X\,\mathrm{d}(X) + Y\,\mathrm{d}(Y) + Z\,\mathrm{d}(Z) + S\,\mathrm{d}(S)}{S^2 + X^2 + Y^2 + Z^2}$$

Note that the Big Omega term is a perfect differential, and is zero only when the arguement R4 is a constant. That is - off the sphere R4 = constant the Omega term does not vanish. Hence if the radius of the two sphere is expanding, then R4 is not constant and one has a dilatation. (The source of dilatons?)

There are in general two sets of torsion two forms.
The affine two forms, big Sigma, which depend upon the product of little omega and the connection components, little gamma.
The second set of torsion 2-forms is related to Big Omega and the connection components, little gamma.

See http://www.uh.edu/~rkiehn/pdf/defects2.pdf

AFFINE TORSION (DISLOCATION) 2-forms Big Sigma

```
> Sigma1:=(wcollect(factor(omega&^gg1)));Sigma2:=wcollect(factor(omega&^gg2));Sigma3:=omega&^gg3;
```

$$\Sigma 1 := -\frac{(Y\,Z+S\,X)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Y)))}{S^2+X^2+Y^2+Z^2}-\frac{(Y\,Z+S\,X)\,((\mathrm{d}(Z))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}-\frac{(-Y^2+S^2)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Z)))}{S^2+X^2+Y^2+Z^2}-\frac{(Z\,S+X\,Y)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(Z)))}{S^2+X^2+Y^2+Z^2}-\frac{(-Z^2+X^2)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}-\frac{(-Z\,S-X\,Y)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}$$

$$\Sigma 2 := -\frac{(-S\,Y+X\,Z)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Y)))}{S^2+X^2+Y^2+Z^2}-\frac{(-S\,Y+X\,Z)\,((\mathrm{d}(Z))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}-\frac{(Z\,S-X\,Y)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Z)))}{S^2+X^2+Y^2+Z^2}-\frac{(-S^2+X^2)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(Z)))}{S^2+X^2+Y^2+Z^2}-\frac{(Z\,S-X\,Y)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}-\frac{(-Z^2+Y^2)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}$$

$$\Sigma 3 := -2\,\frac{Y\,S\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Z)))}{S^2+X^2+Y^2+Z^2}+2\,\frac{Y\,Z\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}+2\,\frac{X\,S\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(Z)))}{S^2+X^2+Y^2+Z^2}-2\,\frac{X\,Z\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(S)))}{S^2+X^2+Y^2+Z^2}$$

```
>
```

DISCLINATION TORSION FORMS

```
> Phi1:=wcollect(factor(Omega&^gg1));Phi2:=wcollect(factor(Omega&^gg2));Phi3:=wcollect(factor(Omega&^gg3));
```

$$\Phi 1 := -\frac{(-X\,Z+S\,Y)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Y)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(-S\,Y+X\,Z)\,((\mathrm{d}(Z))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(Z\,S+X\,Y)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Z)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(Z^2+Y^2)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(Z)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(Z\,S+X\,Y)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(S^2+X^2)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}$$

$$\Phi 2 := -\frac{(Y\,Z+S\,X)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Y)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(-Y\,Z-S\,X)\,((\mathrm{d}(Z))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}$$

$$
\begin{aligned}
\Phi3 := {} & -\frac{(Z^2+X^2)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Z)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(-Z\,S+X\,Y)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(Z)))}{(S^2+X^2+Y^2+Z^2)^2} \\
& -\frac{(-S^2-Y^2)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}-\frac{(Z\,S-X\,Y)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2} \\
& \frac{(-Y^2-X^2)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Y)))}{(S^2+X^2+Y^2+Z^2)^2}+\frac{(-Z^2-S^2)\,((\mathrm{d}(Z))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2} \\
& +\frac{(-Y\,Z+S\,X)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(Z)))}{(S^2+X^2+Y^2+Z^2)^2}+\frac{(X\,Z+S\,Y)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(Z)))}{(S^2+X^2+Y^2+Z^2)^2} \\
& +\frac{(-Y\,Z+S\,X)\,((\mathrm{d}(Y))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}+\frac{(-X\,Z-S\,Y)\,((\mathrm{d}(X))\,\&\wedge\,(\mathrm{d}(S)))}{(S^2+X^2+Y^2+Z^2)^2}
\end{aligned}
$$

What is important is theat the new structural equation related to disclinations generates non-zero values for the second set of torsion 2-forms, in this case named Phi.

Next, the Algebraic method will be used to compute the Matrix of Curvature 2-forms.

(Soon)

>

>

>